```
FILE: MORE      FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE

C           DATA SET MORE        AT LEVEL 001 AS OF 13/03/87
C*****************************************************************
C
C                                MORE
C
C                    MACROPHYTES OCCUPYING RESEVOIRS
C
C            A MODEL FOR SIMULATION OF FLOATING MACROPHYTES ON
C                          (MAN-MADE) LAKES
C
C*****************************************************************
C
C       DEVELOPED BY:          J.G.C. SMITS,
C                              DELFT HYDRAULICS,
C                              P.O. BOX 177, 2600 MH DELFT,
C                              THE NETHERLANDS.
C
C                              PHONE 015-569353
C                              TELEX 38176 HYDEL NL
C
C*****************************************************************
C
C       UNITS FOR INPUT/OUTPUT:
C
C                  4 - INPUT OF PARAMETERS
C                  3 - INPUT OF FORCING FUNCTIONS (HYDROL., HARVESTING)
C                  2 - INPUT OF FORCING FUNCTIONS (METEOR., NUTRIENTS)
C                  6 - OUTPUT OF INPUT
C                  8 - OUTPUT OF MACROPHYTE CONCENTRATIONS
C
C*****************************************************************
C------------------------------------------------------ DECLARATIONS ---
C
      DIMENSION  CB(3),CBO(3),CBI(3),CBD(3),CBW(3),TB(3),TBW(3),PC(3),
     1           FDENS(3),RPROD(3),RMORT(3),RWDES(3),RCROM(3),RHARV(3),
     1           DL(52),WRCO(3),A(3),AO(3),FA(3),FH(3),
     1           TEX(100),TEXT(20,10)
      REAL*4     MCO1,MCO2,MCO3,KSN,KSP,LENS1,LAT
C
C ------------------------------------------------------- SET VALUES ---
C
      NERROR = 0
C
C ------------------------------------------------- READ INPUT DATA ---
C
      READ(4,10) (TEX(I),I= 1,20)
      READ(4,10) (TEX(I),I=21,40)
      READ(4,10) (TEX(I),I=41,60)
      READ(4,10) (TEX(I),I=61,80)
      READ(4,30) (TEXT( 1,I),I=1,8),PGMAX,RDWC,RWDW
      READ(4,30) (TEXT( 2,I),I=1,8),KSN,KSP
      READ(4,30) (TEXT( 3,I),I=1,8),TOPT,TCO1,TCO2
      READ(4,30) (TEXT( 4,I),I=1,8),SIS
      READ(4,30) (TEXT( 5,I),I=1,8),CBMAX,DCO,CBMCOV
      READ(4,30) (TEXT( 6,I),I=1,8),RPWS,WCO1,WCO2,HP
```

```
FILE: MORE     FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE

      READ(4,30) (TEXT( 7,I),I=1,8),PHOPT,PHCO
      READ(4,30) (TEXT( 8,I),I=1,8),RCO1,RCO2,RCO3
      READ(4,30) (TEXT( 9,I),I=1,8),MCO1,MCO2,MCO3
      READ(4,30) (TEXT(10,I),I=1,8),AMAX,HMAX,GAMMA,EN
      READ(4,30) (TEXT(11,I),I=1,8),WIDS1,LENS1,DEPRES,DEPRIV
      READ(4,30) (TEXT(12,I),I=1,8),FA(1),FA(2),FA(3)
      READ(4,30) (TEXT(13,I),I=1,8),WRCO(2),WRCO(3)
      READ(4,30) (TEXT(14,I),I=1,8),CBI(1),CBI(2),CBI(3),ENT
      READ(4,30) (TEXT(15,I),I=1,8),PH,DTEMP,LAT
      READ(4,35) (TEXT(16,I),I=1,8),IOPG,NWB,NWE
      READ(4,30) (TEXT(17,I),I=1,8),DTIME
   10 FORMAT (20A4)
   30 FORMAT (8A4,5F7.0)
   35 FORMAT (8A4,3I7)
C
C------------------------------------------ WRITE INPUT DATA ---
C
      WRITE(6,40)
      WRITE(6,20) (TEX(I),I= 1,20)
      WRITE(6,20) (TEX(I),I=21,40)
      WRITE(6,20) (TEX(I),I=41,60)
      WRITE(6,20) (TEX(I),I=61,80)
      WRITE(6,50) (TEXT( 1,I),I=1,8),PGMAX,RDWC,RWDW
      WRITE(6,50) (TEXT( 2,I),I=1,8),KSN,KSP
      WRITE(6,50) (TEXT( 3,I),I=1,8),TOPT,TCO1,TCO2
      WRITE(6,50) (TEXT( 4,I),I=1,8),SIS
      WRITE(6,50) (TEXT( 5,I),I=1,8),CBMAX,DCO,CBMCOV
      WRITE(6,50) (TEXT( 6,I),I=1,8),RPWS,WCO1,WCO2,HP
      WRITE(6,50) (TEXT( 7,I),I=1,8),PHOPT,PHCO
      WRITE(6,50) (TEXT( 8,I),I=1,8),RCO1,RCO2,RCO3
      WRITE(6,50) (TEXT( 9,I),I=1,8),MCO1,MCO2,MCO3
      WRITE(6,50) (TEXT(10,I),I=1,8),AMAX,HMAX,GAMMA,EN
      WRITE(6,50) (TEXT(11,I),I=1,8),WIDS1,LENS1,DEPRES,DEPRIV
      WRITE(6,50) (TEXT(12,I),I=1,8),FA(1),FA(2),FA(3)
      WRITE(6,50) (TEXT(13,I),I=1,8),WRCO(2),WRCO(3)
      WRITE(6,50) (TEXT(14,I),I=1,8),CBI(1),CBI(2),CBI(3),ENT
      WRITE(6,50) (TEXT(15,I),I=1,8),PH,DTEMP,LAT
      WRITE(6,55) (TEXT(16,I),I=1,8),IOPG,NWB,NWE
      WRITE(6,50) (TEXT(17,I),I=1,8),DTIME
      WRITE(6,45)
      WRITE(8,60)
   20 FORMAT (1H ,20A4)
   40 FORMAT (1H ,'INPUT PARAMETERS:',/,)
   45 FORMAT (1H1,'INPUT FORCING FUNCTIONS:',/,)
   50 FORMAT (1H ,8A4,5F8.3)
   55 FORMAT (1H ,8A4,3I8)
   60 FORMAT (1H ,'DATE      CB1     CBW1       TBW1    PC1      CB2',
     1         '    CBW2       TBW2    PC2      CB3     CBW3       TBW3',
     1         '   PC3',/,)
C
C------------------------------------- CHECKS ON INPUT PARAMETERS ---
C
      IF (DTIME  .LT. 0.001) GOTO 9995
      IF (DEPRIV .LT.   0.1) GOTO 9996
      IF (DEPRES .LT.   0.1) GOTO 9996
```

```
      IF (SIS    .LT.   1.0) GOTO 9996
      IF (HP     .LT. 0.001) GOTO 9996
      IF (CBMCOV .LT.   0.1) GOTO 9996
      DO 65 I=1,3
      IF (FA(I)  .LT. 0.001) GOTO 9996
   65 CONTINUE
C
C------------------------------------------------------ INITIALISATION ---
C
      NC    = 0
      NTS   = IFIX(7.0/DTIME)
      PI    = 3.1415926
      RAD   = PI/180.
      AMAX  = AMAX*1000000.
      WIDS1 = WIDS1*1000.
      LENS1 = LENS1*1000.
      DH    = 0.0
      WRCO(1) = 1.0
      DO 67 I=1,3
      CBO(I) = CBI(I)
   67 CONTINUE
C
C---------- COMPUTE THE DAYLENGTH FOR EACH WEEK AT A GIVEN LATITUDE ---
C---------- DAYLENGTH BECOMES RELEVANT IF TO PRODUCTION IF FIRR IS ---
C------------------------- A NON-LINEAR FUNCTION OF THE INSOLATION ---
C
      DO 70 I=1,52
      DEC   = -ASIN(SIN(23.45*RAD) * COS(2.*PI*(I+1)/52.))
      COSLD = COS(DEC)*COS(LAT*RAD)
      SINLD = SIN(DEC)*SIN(LAT*RAD)
      AOB   = SINLD/COSLD
      DL(I) = 12.*(1.+2.*ASIN(AOB)/PI)
   70 CONTINUE
C
C------------------------------------------------ BEGIN OF WEEKLY LOOP ---
C---------------------------------------- READ FORCING FUNCTIONS ---
C
   80 CONTINUE
      READ(2,100) NYEAR,NWEEK,T,W,WD,SI,CN,CP
  100 FORMAT (2I4,6F8.0)
      READ(3,200) LYEAR,LWEEK,QI,QO,H,(FH(I),I=1,3)
  200 FORMAT (2I4,6F8.0)
      IF (NYEAR  .LT. 1 .OR. LYEAR .LT. 1) GOTO 9999
      NDATE = (NYEAR-1900)*100+NWEEK
C
C---------------------------- WRITE FORCING FUNCTIONS FOR EACH WEEK ---
C
      WRITE(6,220) NDATE,T,W,WD,SI,DL(NWEEK),CN,CP,QI,QO,H,
     1             (FH(I),I=1,3)
  220 FORMAT (1H ,I4,13F8.2)
      IF (NYEAR  .LT. 1900 .OR. NYEAR .NE. LYEAR .OR.
     1    NWEEK .NE. LWEEK) GOTO 9997
      W  = W*86400.
      QI = QI*86400.
      QO = QO*86400.
```

```
      T  = T+DTEMP
C
C-------- DETERMINE IF IT IS GROWING SEASON OR NOT (OPTIONAL, IOPG) ---
C---------------------------------- IF IGS = 0, THE PRODUCTION IS ZERO ---
C
      IGS = 1
      IF (IOPG .EQ. 0)  GOTO 240
      IF (NWB .GT. NWE) GOTO 230
      IF (NWEEK .LT. NWB .OR. NWEEK .GT. NWE) IGS = 0
      GOTO 240
  230 CONTINUE
      IF (NWEEK .LT. NWB .AND. NWEEK .GT. NWE) IGS = 0
  240 CONTINUE
C
C---------------------------------- INITIALISATION OF THE GEOMETRY ---
C
      NC = NC+1
      IF (NC .GT. 1) DH = H-HO
      IF (NC .GT. 1) GOTO 260
      HO = H
      AT = AMAX*(1.-GAMMA*(1.-H/HMAX))**EN
      DO 250 I=1,3
      A(I) = FA(I)*AT
  250 CONTINUE
  260 CONTINUE
C
C---------------------------------- COMPUTE TRANSPORT COEFFICIENTS ---
C
      ALPHA = ABS(WD)
      RENT  = QI*ENT/DEPRIV
      ROUT  = QO/DEPRES + RPWS*W*COS(ALPHA*RAD)*WIDS1
      IF (ROUT .LT. 0.0) ROUT = 0.0
      TROS2 = 0.0
      TROS3 = 0.0
      TRIS2 = 0.0
      TRIS3 = 0.0
      TROS1 = RPWS*W*SIN(ALPHA*RAD)*LENS1
      IF (WD .GT. 0.0) GOTO 300
      TROS2 = WRCO(2)*TROS1
      TRIS3 = TROS1
      GOTO 310
  300 TRIS2 = TROS1
      TROS3 = WRCO(3)*TROS1
  310 CONTINUE
C
C---------------------------- COMPUTE CONSTANT LIMITING FUNCTIONS ---
C
      IF (T .GT. TOPT) GOTO 350
      FTEMP = EXP(-TCO1*((TOPT-T)**2))
      GOTO 360
  350 FTEMP = EXP(-TCO2*((TOPT-T)**2))
  360 CONTINUE
      FNIT  = CN/(KSN+CN)
      FPHOS = CP/(KSP+CP)
      FNUT  = FNIT
```

```
FILE: MORE     FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE


      IF (FPHOS .LT. FNIT) FNUT = FPHOS
      FIRR  = SI/SIS
      IF (SI .GT. SIS) FIRR = 1.0
      FPH   = 1.0 - PHCO*((PH-PHOPT)**2)
C
C------------------------------------ COMPUTE CONSTANT LOSS RATES ---
C
      DO 370 I=1,3
      HW = WCO1*WRCO(I)*W/86400.
      RWDES(I) = WCO2*HW/HP
      RHARV(I) = FH(I)/7.0
  370 CONTINUE
C
C--------------------------- BEGIN WEEKLY LOOP FOR TIMESTEP DTIME ---
C
      DO 550 NS=1,NTS
C
C---------------------------------- DETERMINATION OF THE GEOMETRY ---
C
      IF (DH .EQ. 0.0) GOTO 395
      HO  = HO + DH*DTIME/7.0
      ATO = AT
      AT  = AMAX*(1.-GAMMA*(1.-HO/HMAX))**EN
      DO 380 I=1,3
      AO(I) = A(I)
      A(I)  = FA(I)*AT
  380 CONTINUE
C
C------- COMPUTE NEW BIOMASS DENSITY IN RELATION TO CHANGE IN LEVEL ---
C
      IF (DH .LT. 0.0 .AND. ATO-AT .GT. AO(2)+AO(3)) GOTO 9998
      IF (DH .GT. 0.0 .AND. AT-ATO .GT. AO(2)+AO(3)) GOTO 9998
      F2 = FA(2)/(FA(2)+FA(3))
      F3 = FA(3)/(FA(2)+FA(3))
      IF (DH .GT. 0.0) GOTO 390
      CB(2) = (CB(2)*AO(2) + F2*(AO(1)-A(1))*CB(1) -
     1         F2*(ATO-AT)*CB(2))/A(2)
      CB(3) = (CB(3)*AO(3) + F3*(AO(1)-A(1))*CB(1) -
     1         F3*(ATO-AT)*CB(3))/A(3)
      CB(1) = CB(1)
      GOTO 395
  390 CONTINUE
      CB(2) = (CB(2)*AO(2) - F2*(A(1)-AO(1))*CB(2))/A(2)
      CB(3) = (CB(3)*AO(3) - F3*(A(1)-AO(1))*CB(3))/A(3)
      CB(1) = (CB(1)*AO(1) + F2*(A(1)-AO(1))*CB(2) +
     1         F3*(A(1)-AO(1))*CB(3))/A(1)
  395 CONTINUE
C
C------------ COMPUTE NEW BIOMASS DENSITY IN RELATION TO TRANSPORT ---
C
      CB(1) = CBO(1) + (RENT - ROUT*CBO(1) + TROS2*CBO(2) +
     1                  TROS3*CBO(3) - TROS1*CBO(1))*DTIME/A(1)
      CB(2) = CBO(2) + (TRIS2*CBO(1) - TROS2*CBO(2))*DTIME/A(2)
      CB(3) = CBO(3) + (TRIS3*CBO(1) - TROS3*CBO(3))*DTIME/A(3)
C
```

```
FILE: MORE     FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE

C------------------------------ COMPUTE CHANGING LIMITING FUNCTIONS ---
C
      DO 400 I=1,3
      FDENS(I) = (DCO+1)/(CBO(I)/(DCO*CBMAX)+1) - DCO
  400 CONTINUE
      DO 440 I=1,3
      RCROM(I) = MCO1*(CBO(I)-CBMCOV)/CBO(I)
      IF (RCROM(I) .LT. 0.0) RCROM(I) = 0.0
  440 CONTINUE
C
C------------------------------ COMPUTE PRODUCTION AND LOSS RATES ---
C
      DO 450 I=1,3
      RPROD(I) = PGMAX*FTEMP*FNUT*FIRR*FPH*FDENS(I)*(1.-RCC1) -
     1           EXP(RCO2*T-RCO3)
      RMORT(I) = EXP(MCO2*T-MCO3) + RCROM(I) + RWDES(I)
  450 CONTINUE
C
C------------- COMPUTE NEW BIOMASS DENSITY IN RELATION TO PROCESSES ---
C
      DO 500 I=1,3
      IF (CB(I) .LT. 0.0) CB(I) = 0.0
      CB(I)  = CB(I) + (RPROD(I)-RMORT(I)-RHARV(I))*CBO(I)*DTIME
      CBO(I) = CB(I)
  500 CONTINUE
  550 CONTINUE
      HO  = H
      ATO = AT
C
C------------------------------------------------- PREPARE OUTPUT ---
C
      DO 800 I=1,3
      CBD(I) = CB(I)*RDWC
      CBW(I) = CBD(I)*RWDW
      TB(I)  = CB(I)*A(I)/1000000.
      TBW(I) = TB(I)*RDWC*RWDW
      PC(I)  = 100.*(1.0-(CBMCOV-CB(I))/CBMCOV)
      IF (CB(I) .GT. CBMCOV) PC(I) = 100.0
  800 CONTINUE
C
C------------------------------------------------- WRITE OUTPUT ---
C
      WRITE(8,850) NDATE,(CB(I),CBW(I),TBW(I),PC(I),I=1,3)
  850 FORMAT (1H ,I4,4(2F8.2,F10.0,F6.0))
C
C------------------------------------ END OF THE WEEKLY LOOP ---
C
      GOTO 80
C
C------------------------------------------ END OF PROGRAM ---
C
 9995 NERROR = NERROR+1
 9996 NERROR = NERROR+1
 9997 NERROR = NERROR+1
 9998 NERROR = NERROR+1
```

```
FILE: MORE     FORTRAN  A1  VM/SP HPO REL 4.2 - ELETRONORTE

      WRITE(6,900) NERROR
  900 FORMAT (1H ,'NERROR =',I2)
 9999 CONTINUE
      STOP
      END
```